/**
 * Copyright (c) 2007-2008, the original author(s)
 * 
 * All rights reserved.
 * 
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 * 
 *     * Redistributions of source code must retain the above copyright notice,
 *       this list of conditions and the following disclaimer.
 *     * Redistributions in binary form must reproduce the above copyright
 *       notice, this list of conditions and the following disclaimer in the
 *       documentation and/or other materials provided with the distribution.
 *     * Neither the name of the Prana Framework nor the names of its contributors
 *       may be used to endorse or promote products derived from this software
 *       without specific prior written permission.
 * 
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS
 * "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT
 * LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR
 * A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT OWNER OR
 * CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL,
 * EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO,
 * PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR
 * PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF
 * LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING
 * NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS
 * SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.
 */
 package org.pranaframework.utils {
	
 	/**
 	 * Converts a variable from one type to another.
 	 * 
 	 * @author Christophe Herreman
 	 */
	public class TypeConverter {
		
		/**
		 * Converts a variable from a String to the best suited type for the
		 * variable.
		 * 
		 * @param value the value to convert
		 */
		public static function execute(value:String):* {
			var result:*;
			switch (true) {
				case "true" == value:
					result = true;
					break;
				case "false" == value:
					result = false;
					break;
				case !isNaN(Number(value)):
					result = Number(value);
					break;
				default:
					// check for a class
					try {
						result = ClassUtils.forName(value);
					}
					catch (e:Error) {
						// this is not a class, let's look for a static property
						if (value.indexOf(".") != -1) {
							var a:Array = value.split(".");
							var last:String = a.pop() as String;
							try {
								result = ClassUtils.forName(a.join("."));
								if (result[last]) {
									result = result[last];
								}
							}
							catch (e:Error) {
								result = value;
							}
						}
						else {
							result = value;
						}
					}
					
					
					break;
			}
			return result;
		}
	}
}